


Math 2280-2
Tuesday April 17
Fourier Series and the Heat Equation

These commands will be useful for the first part of your final 2280 Maple assignment - which is to do the section 9.5 computing project. The second part of your assignment is to do the section 9.6 computing project. You may make your own solution files.

We will also use this output in class today, April 17.

```
> restart:with(plots):
Warning, the name changecoords has been redefined
> a:=n->2/L*int(f(x)*cos(Pi*n/L*x),x=0..L):
    #for a cosine series! Now use x because it will
    #be space variable
b:=n->2/L*int(f(x)*sin(Pi*n/L*x),x=0..L):
    #for a sine series!
> cossum:=a(0)/2 + sum(a(n)*cos(Pi*n/L*x),n=1..N):
    #cosine series
sinsum:=sum(b(n)*sin(Pi*n/L*x),n=1..N):
    #sine series
> heatsol1:=sum(b(n)*sin(Pi*n/L*x)*exp(-n^2*Pi^2*k*t/L^2),n=1..N):
    #boundary conditions u(0,t)=u(L,t)=0
heatsol2:=a(0)/2 +
    sum(a(n)*cos(Pi*n/L*x)*exp(-n^2*Pi^2*k*t/L^2),n=1..N):
    #insulated end boundary conditions
> #Example 2 page 636
f:=x->100:
L:=50:
N:=50: #take many terms because the square wave
    #is hard to approximate
k:=0.15: #iron
> plot3d(heatsol1,x=0..50,t=0..1800,color=black,
    style=wireframe, title="Figure 9.5.4");
```

Figure 9.5.4



Computer check of computations on page 637

(a) temperature at midpoint after 1800 seconds (= 6 minutes), for iron


```
[ > evalf(subs({t=1800,x=25},heatsol1));
                                     43.84897699
```

(b) temperature for concrete:

```
[ > k:=.005:
  evalf(subs({t=1800,x=25},heatsol1));
                                     99.99999917
[ > evalf(subs({t=60*60,x=25},heatsol1));
  #an hour later
                                     99.99381821
[ > evalf(subs({t=6*60*60,x=25},heatsol1));
  #6 hours later - think of brick buildings!
                                     82.21276660
```

We could see what the temperature looked like throughout the concrete slab at this last time value:


```
[ > slice:=z->evalf(subs({t=6*60*60,x=z},heatsol1));
                                     slice:=z->evalf(subs({x=z,t=21600},heatsol1))
[ > plot(slice(x),x=0..50,color=black,
  title="temperature profile six hours later, concrete");
```



So, if you wanted to, you could also trace the temperature at the midpoint, as time varies.....


Example 3, page 640:

```
> f:=x->4*(x -2*(Heaviside(x-25))*(x-25)):
  plot(f(x),x=0..50,color=black);
```



```
> k:=0.15:
  #back to iron
> plot3d(heatsol2,x=0..50,t=0..1200,color=black,
  style=wireframe, axes=boxed, title= "Figure 9.5.6");
```

Figure 9.5.6



[>